United States
Department of
Agriculture

Forest
Service

North Central
Forest Experiment
Station

Research
Paper NC-207

GROW— a Computer Subroutine

that Projects the Growth of Trees in Lake States' Forests

Gary J. Brand



North Central Forest Experiment Station
Forest Service—U.S. Department of Agriculture
1992 Folwell Avenue
St. Paul, Minnesota 55108

Manuscript approved for publication September 29, 1980
1981

"GROW"—A COMPUTER SUBROUTINE THAT PROJECTS THE GROWTH OF TREES IN LAKE STATES' FORESTS

Gary J. Brand, *Research Forester*

STEMS (Stand and Tree Evaluation and Modeling System) is comprised of two computer programs that simulate the growth, death, cutting, and regeneration of trees in Lake States' (Michigan, Minnesota, and Wisconsin) forests.[1] STEMS projects forest stands containing any mixture of tree species and sizes. The system was also designed to be flexible in the type of summary information it could provide. This flexibility is achieved by maintaining a tree list that describes the stand being projected. Examples of the type of summaries that can be produced from the tree lists are stand tables, stock tables, and yield tables.

The first program simulates growth, death, cutting, and regeneration. Its core component is a distance-independent, individual tree growth model that simulates both tree growth and mortality.[2] The second program summarizes the results of the projection in a variety of user-selectable ways.

Although STEMS can be used in many ways, a simpler system would be desirable in some situations. A small growth model could be an integral part of an analysis program whereas STEMS is too large to be included within another program. Secondly, a small model containing only the parts essential to growth would be easier to understand and evaluate. Thirdly, a smaller system would be cheaper to use.

I have developed a simpler version of STEMS called Grow that contains only the growth and mortality parts of STEMS. Grow contains two subprograms (fig. 1). The first is a subroutine that grows and kills trees. The second is a BLOCK DATA subprogram that contains all the coefficients for the functions in the model. Like STEMS, Grow will simulate growth and mortality for any size or species mixture encountered in the Lake States. Unlike STEMS, it does not simulate cutting or regeneration, read initial conditions, or produce summary tables. Therefore, the user must develop a main or calling program to read initial conditions, call Grow, and produce summaries.

To implement Grow the user must provide a tree list representing the stand to be projected and a computer program to read this list. A common tree list is a forest inventory plot from which tree species and diameters have been recorded. The tree characteristics required are d.b.h., species group code, status code, tree factor (the number of trees per acre the tree represents), and crown ratio code.[3] The stand site index is also required. Any tree list can be used as long as it has the information required.

[1] *STEMS is the latest version of a series of computer programs designed to implement a generalized tree growth projection system. It was developed by a team of researchers at the North Central Forest Experiment Station. Included in the team were Jerold T. Hahn, David M. Belcher, Gary J. Brand, Roland G. Buchman, Stephen E. Fairweather, Margaret R. Holdaway, and Stephen R. Shifley. A description of STEMS is being written by David M. Belcher, Margaret R. Holdaway, and Gary J. Brand.*

[2] *The original growth model (U.S. Department of Agriculture, Forest Service. 1979. A generalized forest growth projection system applied to the Lake States Region. U.S. Department of Agriculture Forest Service, General Technical Report NC-49, 96 p. U.S. Department of Agriculture Forest Service, North Central Forest Experiment Station, St. Paul, Minnesota). was developed by a team of researchers at the North Central Forest Experiment Station. The team was led by Dr. Rolfe A. Leary and included Roland G. Buchman, Jerold T. Hahn, and Margaret R. Holdaway. This original model has since been modified and it is this modified version that is used in STEMS and Grow.*

[3] *Crown ratio code indicates the percent of the tree's total height that is in full live crown. A code of 1 means that between 1 and 10 percent of the total height is full live crown. Each additional 10 percent of full live crown increases the code by 1. Grow will calculate a crown ratio code for trees without a code using a function developed by Margaret R. Holdaway of the North Central Forest Experiment Station, St. Paul, Minnesota. This function is also used to produce a crown ratio code for each subsequent year of the projection.*



Figure 1.—Flow diagram of Grow.

The user's program must read tree characteristics into five arrays dimensioned to the longest expected tree list. Elements 1,2,...,n of each array correspond to characteristics of trees 1,2,...,n, where n is the number of trees in the list. A sixth array, dimensioned the same as the previous arrays, is needed to hold a crown ratio code correction. The correction is the difference between the observed and the predicted crown ratio code and is calculated by Grow. This correction is added to each subsequent calculation of crown ratio for each tree. Once the tree list is read into the arrays, the user selects the number of years the stand will be projected and the type of mortality desired.

To project the growth of the stand the user-written program must call subroutine Grow. The call contains the following variables (in this order):

number of years to project the stand,
number of trees in the tree list,
d.b.h. array,
species group code array,
status code array,
tree factor array,
crown ratio code array,
crown ratio code correction array,
stand site index (in feet, base age 50 years),
mortality code,
new plot flag (indicates this is the first call to Grow for the plot), and
error flag.

A more complete description of these variables is given in the Appendix.

Two types of mortality—deterministic and stochastic—can be calculated by Grow. The same mortality function is used for each type but it is applied differently. Deterministic mortality is indicated by a mortality code of 0. This type of mortality annually reduces the tree factor of a tree by an amount proportional to its calculated annual mortality rate. For example, if the tree factor was 10 trees per acre and the mortality rate was 0.05, the new tree factor would be 9.5 trees per acre ($10 \times (1-0.05)$). Stochastic mortality is indicated by a mortality code of 1. With this option the calculated mortality rate is compared with a random number drawn from a uniform distribution from 0 to 1. If the random number is less than the mortality rate, the status code is set to dead. Stochastic mortality is also used for trees that have tree factors less than 1 when deterministic mortality is used.

Grow returns a projected tree list with updated diameters, status codes, tree factors, and crown ratio codes. These updated characteristics are contained in the same arrays that were passed to Grow. For instance, the new d.b.h. for the first tree is in the first position of the diameter array. The user must provide the appropriate computer statements to summarize the new tree list. If summaries of volume are desired, local volume equations must be incorporated. Repeated calls to Grow will produce additional projections of the stand.

To compare their storage requirements and costs, I projected a sample plot using both STEMS and Grow on a CDC Cyber 74 computer. Grow required 35 percent of the computer storage required by STEMS and the computing cost for Grow was 17 percent of the cost for STEMS. The larger program, however, produced both yield and components of growth summaries for basal area, number of trees, and volume. Grow produced only basal area and number of trees yield summaries. These differences may vary on other computer systems.

Grow provides a simple and inexpensive way to use the STEMS Lake States' growth model. To use Grow the user must write a program to read tree lists, call Grow, and summarize the results. A copy of Grow, a sample calling program, and a sample tree list punched on computer cards can be obtained from the author.

APPENDIX

```
      SUBROUTINE GROW (NYRS,NT,DBH,ISPG,ISTAT,TRFAC,CRC,CRD,
     1 SI,MORT,NUPL,IERR)
C
C
C                              WRITTEN BY:
C
C                       GARY BRAND
C                       NORTH CENTRAL FOREST EXPERIMENT STATION
C                       1992 FOLWELL AVENUE
C                       ST.  PAUL, MN 55108
C                       612-642-5276
C                       JUNE 10,1980
C
C  TO USE THIS SUBROUTINE THE FOLLOWING VARIABLES MUST BE SET IN
C     THE CALLING PROGRAM AND PASSED THROUGH THE CALL STATEMENT:
C
C   NYRS    - NUMBER OF YEARS IN PROJECTION
C   NT      - NUMBER OF TREES IN TREE LIST
C   DBH     - ARRAY OF DBH'S OF TREES
C   ISPG    - ARRAY OF SPECIES GROUP CODES OF TREES
C      1-JACK PINE       11-BLACK ASH      21-N.  RED OAK
C      2-RED PINE        12-COTTONWOOD     22-OTHER RED OAK
C      3-WHITE PINE      13-SILVER MAPLE   23-HICKORY
C      4-WHITE SPRUCE    14-RED MAPLE      24-BIGTOOTH ASPEN
C      5-BALSAM FIR      15-ELM            25-QUAKING ASPEN
C      6-BLACK SPRUCE    16-YELLOW BIRCH   26-PAPER BIRCH
C      7-TAMARACK        17-BASSWOOD       27-RED PINE PLANTATION
C      8-N. WHITE-CEDAR  18-HARD MAPLE     28-29 NOT USED
C      9-HEMLOCK         19-WHITE ASH      30-OTHER HARDWOODS
C     10-OTHER SOFT.     20-WHITE OAK      31-NONCOMMERCIAL
C
C   ISTAT   - ARRAY OF STATUS CODES OF TREES ( 1=LIVE 2=CUT 3=DEAD )
C   TRFAC   - ARRAY OF TREE FACTORS OF TREES (TREES/ACRE REPRESENTED BY
C              EACH TREE IN LIST)
C   CRC     - ARRAY OF CROWN RATIO CODES OF TREES (0-10%=1, 11-20%=2,..
C              81-90%=9, 91-100%=10)
C   CRD     - ARRAY OF DIFFERENCES BETWEEN OBSERVED AND PREDICTED CROWN
C              RATIOS OF TREES - RETURNED BY GROW
C   SI      - SITE INDEX (HEIGHT AT 50 YEARS)
C   MORT    - SELECTS TYPE OF MORTALITY,
C             MORT=0 DETERMINISTIC
C             MORT=1 STOCHASTIC
C   NUPL    - NEW PLOT INDICATOR SET TO ZERO BY THE CALLING PROGRAM WHEN
C              A NEW PLOT IS ENTERED
C   IERR    - ERROR FLAG RETURNED BY GROW ,
C              IERR=0 IF NO ERROR
C              IERR=1 IF NYRS IS LESS THAN 1
C              IERR=2 IF NT IS LESS THAN 1
C
C   PROJECTED TREE ATTRIBUTES ARE RETURNED AS THE SAME VARIABLES
C      THAT WERE SENT IN THE CALL
C
C   DBH, ISPG, ISTAT, TRFAC, CRC, AND CRD MUST BE DIMENSIONED IN THE
C      CALLING PROGRAM
```

```
C
C
      PARAMETER NSG=31
      COMMON /GROPAR/ CRP(4,NSG),PGP(5,NSG),PMP(5,NSG),GMPO(4,NSG),
     1 GMPB(2,NSG),GMPC(2,NSG)
      DIMENSION DBH(NT),ISPG(NT),ISTAT(NT),TRFAC(NT),CRC(NT),CRD(NT)
      DIMENSION RUNBA(10)
      NTREES=NT
C
C         NUPL EQUALS 0 WHEN A NEW PLOT IS STARTED
      IF(NUPL.EQ.0) THEN
C
C          CLEAR RUNNING AVERAGE BASAL AREA ARRAY.
        DO 5 I=1,10
    5   RUNBA(I)=0.0
      ENDIF
C
C          DO NOT PROJECT THE PLOT IF THERE ARE NO TREES OR YEARS
      IF(NYRS.LT.1.OR.NTREES.LT.1) THEN
        IF(NYRS.LT.1) IERR=1
        IF(NTREES.LT.1) IERR=2
      ELSE
        DO 400 IYR=1,NYRS
          AVDBH=0.0
          BA=0.0
          ALIVTR=0.0
          DO 10 I=1,NTREES
            IF(ISTAT(I).NE.1) GO TO 10
            IF(DBH(I).LE.0.0) GO TO 10
            ALIVTR=ALIVTR+TRFAC(I)
            AVDBH=AVDBH+DBH(I)*TRFAC(I)
            BA=BA+DBH(I)*DBH(I)*TRFAC(I)
   10     CONTINUE
C
C          CALCULATE RUNNING AVERAGE BA FOR UP TO THE PAST 10 YEARS
          BA=0.005454*BA
          AVDBH=AVDBH/ALIVTR
          BATOT=BA
          JBA=1
          DO 20 I=1,9
            IF(RUNBA(I+1).GT.0.0) THEN
              RUNBA(I)=RUNBA(I+1)
              JBA=JBA+1
              BATOT=BATOT+RUNBA(I)
            ENDIF
   20     CONTINUE
          RUNBA(10)=BA
          AV10BA=BATOT/FLOAT(JBA)
C
C          GROW THE LIVE TREES
          DO 50 I=1,NTREES
            IF(ISTAT(I).NE.1) GO TO 50
            ISG=ISPG(I)
C
```

```
C          CALCULATE THE CROWN RATIO
            CR=CRP(1,ISG)/(1.+CRP(2,ISG)*AV10BA)+CRP(3,ISG)*(1.-
     1        EXP(CRP(4,ISG)*DBH(I)))
            IF(NUPL.EQ.0.AND.IYR.EQ.1) THEN
C
C          CORRECT THE PREDICTED CROWN RATIO ACCORDING TO THE OBSERVED
C            CROWN RATIO AND STORE THE CORRECTION IN CRD
              IF(CRC(I).LE.0.0) CRC(I)=CR
              CRD(I)=CRC(I)-CR
            ELSE
              CRC(I)=CR+CRD(I)
              IF(CRC(I).LT..5) CRC(I)=.5
              IF(CRC(I).GT.10.0) CRC(I)=10.0
            ENDIF
C
C          CALCULATE THE POTENTIAL GROWTH
            PG=PGP(1,ISG)+PGP(2,ISG)*DBH(I)**PGP(3,ISG)+PGP(4,ISG)
     1        *SI*CRC(I)*DBH(I)**PGP(5,ISG)
C
C          CALCULATE THE MODIFIER OF THE POTENTIAL
C             RED PINE PLANTATION COEFFICIENTS ARE STORED IN GROUP 27.
C             IF THESE COEFFICIENTS ARE DESIRED USE JSG=27.
            JSG=ISG
            RATIO=DBH(I)/AVDBH
            OMEGA=GMPO(1,JSG)*(1.-EXP(GMPO(2,JSG)*RATIO))**GMPO(3,JSG)
     1        +GMPO(4,JSG)
            BETA=GMPB(1,JSG)*(AVDBH+1.)**GMPB(2,JSG)
            CHECK=GMPC(2,JSG)-BA
            IF(CHECK.LT.0.0) CHECK=0.0
            GM=GMPC(1,JSG)*(1.0-EXP(-OMEGA*BETA*SQRT(CHECK/BA)))
C
C          CALCULATE THE PREDICTED GROWTH
            DGRO=PG*GM
            IF(DGRO.LT.0.0) DGRO=0.0
            DBH(I)=DBH(I)+DGRO
C
C          CALCULATE THE PREDICTED MORTALITY RATE
            ARG=PMP(1,ISG)+PMP(2,ISG)*DGRO**PMP(3,ISG)+PMP(5,ISG)*
     1        DBH(I)
            IF(ARG.GT.740.2) ARG=740.2
            PM=1.0/(1.0+EXP(ARG))+PMP(4,ISG)
C
C       DRAW A RANDOM NUMBER FROM A UNIFORM DISTRIBUTION  (0 TO 1)
C         TO DETERMINE IF THE TREE WILL DIE (WHEN MORT=1 OR TREE
C         FACTOR IS LESS THAN 1.0)
            IF(MORT.EQ.1.OR.TRFAC(I).LT.1.0) THEN
C
C       IF NECESSARY, REPLACE RANF(0.0) WITH AN APPROPRIATE ROUTINE
C        THAT PRODUCES A RANDOM NUMBER UNIFORMLY DISTRIBUTED
C         BETWEEN 0 AND 1.
              RAN=RANF(0.0)
              IF(RAN.LE.PM) ISTAT(I)=3
C
C       REDUCE THE TREE FACTOR TO ACCOUNT FOR MORTALITY
```

```
         ELSE
           TRFAC(I)=TRFAC(I)*(1.-PM)
         ENDIF
 50    CONTINUE
400   CONTINUE
      IERR=0
     ENDIF
     NUPL=1
     RETURN
     END
     BLOCK DATA COEF
C
C         ALL GROWTH COEFFICIENTS ARE SET IN THIS ROUTINE
     COMMON /GROPAR/ CRP(4,31),PGP(5,31),PMP(5,31),GMPO(4,31),
    1 GMPB(2,31),GMPC(2,31)
C
C          CROWN RATIO COEFFICIENTS
     DATA ((CRP(I,J),I=1,4),J=1,16)/                                  ISPG
    C      .66400E+01  , .01350E+00 ,  .32000E+01  ,-.05180E+00 ,      01 JP
    C      .53500E+01  , .00530E+00 ,  .15280E+01  ,-.03300E+00 ,      02 RP
    C      .67900E+01  , .00580E+00 ,  .75900E+01  ,-.01030E+00 ,      03 WP
    C      .78400E+01  , .00570E+00 ,  .12720E+01  ,-.14200E+00 ,      04 WS
    C      .56300E+01  , .00470E+00 ,  .35230E+01  ,-.06890E+00 ,      05 BF
    C      .55400E+01  , .00720E+00 ,  .42000E+01  ,-.05300E+00 ,      06 BS
    C      .60000E+01  , .00530E+00 ,  .43100E+00  ,-.00120E+00 ,      07 TM
    C      .57100E+01  , .00770E+00 ,  .22900E+01  ,-.25300E+00 ,      08 NWC
    C      .57100E+01  , .00770E+00 ,  .22900E+01  ,-.25300E+00 ,      09 HEM
    C      .57100E+01  , .00770E+00 ,  .22900E+01  ,-.25300E+00 ,      10 OS
    C      .45000E+01  , .00320E+00 ,  .79500E+00  ,-.10500E+00 ,      11 BA
    C      .43500E+01  , .00460E+00 ,  .18200E+01  ,-.27400E+00 ,      12 COT
    C      .48500E+01  , .00500E+00 ,  .98100E+01  ,-.00990E+00 ,      13 SIL
    C      .43500E+01  , .00460E+00 ,  .18200E+01  ,-.27400E+00 ,      14 RM
    C      .44000E+01  , .00250E+00 ,  .10000E+01  ,-.09400E+00 ,      15 ELM
    C      .41800E+01  , .00250E+00 ,  .14100E+01  ,-.51200E+00 /      16 YB
     DATA ((CRP(I,J),I=1,4),J=17,31)/
    C      .44400E+01  , .00370E+00 ,  .20900E+01  ,-.06500E+00 ,      17 BW
    C      .34000E+01  , .00660E+00 ,  .28700E+01  ,-.43400E+00 ,      18 SM
    C      .44900E+01  , .00290E+00 ,  .12100E+01  ,-.06500E+00 ,      19 WA
    C      .58400E+01  , .00820E+00 ,  .32600E+01  ,-.04900E+00 ,      20 WO
    C      .42000E+01  , .00160E+00 ,  .27600E+01  ,-.02500E+00 ,      21 RO
    C      .50600E+01  , .00330E+00 ,  .17300E+01  ,-.06100E+00 ,      22 ORO
    C      .62100E+01  , .00730E+00 ,  .99900E+01  ,-.01000E+00 ,      23 HIC
    C      .41100E+01  , .00540E+00 ,  .16500E+01  ,-.11000E+00 ,      24 BTA
    C      .40000E+01  , .00240E+00 , -.28300E+01  , .02100E+00 ,      25 QA
    C      .50000E+01  , .00660E+00 ,  .49200E+01  ,-.02630E+00 ,      26 PB
    C      .40000E+01  , .00240E+00 , -.28300E+01  , .02100E+00 ,      27
    C      .40000E+01  , .00240E+00 , -.28300E+01  , .02100E+00 ,      28
    C      .40000E+01  , .00240E+00 , -.28300E+01  , .02100E+00 ,      29
    C      .40000E+01  , .00240E+00 , -.28300E+01  , .02100E+00 ,      30 OH
    C      .40000E+01  , .00240E+00 , -.28300E+01  , .02100E+00 /      31 NC
C
C          POTENTIAL GROWTH COEFFICIENTS
     DATA ((PGP(I,J),I=1,5),J=1,16)/                                  ISPG
    C  .16062E-00,-.90000E-05, .36245E+01, .40000E-04, .10000E+01,     01
```

```
     C  .94460E-01,-.12000E-03, .20596E+01, .35000E-03, .24225E+00,       02
     C  .25578E-00,-.88000E-03, .17263E+01, .40000E-04, .10000E+01,       03
     C  .17056E-00,-.14516E-01, .10660E+01, .52000E-03, .27298E+00,       04
     C  .12200E-00,-.80000E-03, .19890E+01, .60000E-04, .10000E+01,       05
     C  .10713E+00,-.10700E-02, .20017E+01, .60000E-04, .91127E+00,       06
     C  .11147E-00,-.10000E-04, .30685E+01, .30000E-04, .10000E+01,       07
     C  .13403E-00,-.10000E-05, .36880E+01, .20000E-04, .10000E+01,       08
     C  .16872E-00,-.30000E-06, .35738E+01, .10000E-04, .10000E+01,       09
     C  .16062E-00,-.90000E-05, .36245E+01, .40000E-04, .10000E+01,       10
     C  .58807E-01,-.60000E-08, .50559E+01, .24000E-03, .31553E-00,       11
     C  .10948E-00,-.40000E-04, .22226E+01, .50000E-03, .66260E-01,       12
     C  .10948E-00,-.40000E-04, .22226E+01, .50000E-03, .66260E-01,       13
     C  .10948E-00,-.40000E-04, .22226E+01, .50000E-03, .66260E-01,       14
     C  .28496E-00,-.18200E-02, .15297E+01, .30000E-04, .10000E+01,       15
     C  .15155E-00,-.30000E-05, .33104E+01, .70000E-04, .57302E+00/       16
      DATA ((PGP(I,J),I=1,5),J=17,31)/
     C  .25402E-00,-.40000E-05, .29396E+01, .10000E-04, .10000E+01,       17
     C  .18772E-00,-.70000E-05, .25839E+01, .28000E-03, .83850E-01,       18
     C  .21167E-00,-.30000E-05, .33131E+01, .10000E-04, .12430E+00,       19
     C  .12654E-00,-.40000E-05, .27538E+01, .50000E-04, .62086E+00,       20
     C  .15535E-00,-.10000E-06, .35367E+01, .18000E-03, .25900E-00,       21
     C  .17358E-00,-.70000E-04, .24451E+01, .30000E-04, .95222E-00,       22
     C  .16471E-00,-.21610E-02, .13949E+01, .40000E-04, .71628E-00,       23
     C  .23490E-00,-.94200E-02, .11041E+01, .36000E-03, .15386E-00,       24
     C  .21645E-00,-.91000E-04, .26030E+01, .44000E-04, .10000E+01,       25
     C  .10971E-00,-.32000E-03, .20236E+01, .34000E-03, .21288E-00,       26
     C  .37510E-00,-.24790E-01, .96288E+00, .40000E-04, .10000E+01,       27
     C  .37510E-00,-.24790E-01, .96288E+00, .40000E-04, .10000E+01,       28
     C  .37510E-00,-.24790E-01, .96288E+00, .40000E-04, .10000E+01,       29
     C  .37510E-00,-.24790E-01, .96288E+00, .40000E-04, .10000E+01,       30
     C  .37510E-00,-.24790E-01, .96288E+00, .40000E-04, .10000E+01/       31
C
C          MORTALITY COEFFICIENTS
      DATA ((PMP(I,J),I=1,5),J=1,16)/                                    ISPG
     C  .36767E+00, .60178E+02, .10727E+01, .00502E+00, .12220E+00  ,     01
     C  .10843E+01, .11012E+02, .58866E+00, .00020E+00, .48500E+00  ,     02
     C  .23507E+01, .33609E+02, .10088E+01, .00201E+00, .00000E+00  ,     03
     C  .28806E+01, .30774E+02, .12309E+01, .00396E+00, .00000E+00  ,     04
     C  .26976E+01, .30550E+02, .89963E+00, .00400E+00,-.23069E+00  ,     05
     C  .28774E+01, .43295E+03, .18576E+01, .00235E+00, .00000E+00  ,     06
     C  .36767E+00, .60178E+02, .10727E+01, .00502E+00, .12220E+00  ,     07
     C  .28774E+01, .43295E+03, .18576E+01, .00235E+00, .00000E+00  ,     08
     C  .28233E+01, .89560E+02, .10966E+01, .00320E+00, .00000E+00  ,     09
     C  .10843E+01, .11012E+02, .58866E+00, .00020E+00, .48500E+00  ,     10
     C  .14875E+01, .13338E+02, .54290E+00, .00018E+00, .00000E+00  ,     11
     C  .22405E+01, .56594E+02, .10392E+01, .00494E+00, .00000E+00  ,     12
     C  .22405E+01, .56594E+02, .10392E+01, .00494E+00, .00000E+00  ,     13
     C  .22405E+01, .56594E+02, .10392E+01, .00494E+00, .00000E+00  ,     14
     C  .19889E+01, .44541E+02, .10965E+01, .00899E+00, .00000E+00  ,     15
     C  .26705E+01, .63148E+02, .13819E+01, .00382E+00, .00000E+00  /     16
      DATA ((PMP(I,J),I=1,5),J=17,31)/
     C -.58612E+00, .10187E+03, .16493E+01, .00489E+00, .68398E+00  ,     17
     C  .81500E+00, .66937E+02, .15934E+01, .00435E+00, .75000E+00  ,     18
     C  .17821E+01, .23231E+03, .17267E+01, .00222E+00, .00000E+00  ,     19
```

```
     C  .36946E+01, .95637E+02, .14517E+01, .00188E+00, .00000E+00   ,     20
     C  .23092E+01, .69998E+04, .26615E+01, .00468E+00, .00000E+00   ,     21
     C  .23092E+01, .69998E+04, .26615E+01, .00468E+00, .00000E+00   ,     22
     C  .30626E+01, .63495E+02, .10498E+01, .00134E+00, .00000E+00   ,     23
     C  .88050E+00, .56593E+02, .12154E+01, .01288E+00, .00000E+00   ,     24
     C -.07855E+00, .21363E+03, .17126E+01, .02276E+00, .25812E+00   ,     25
     C  .21808E+01, .11626E+02, .47500E+00, .00029E+00, .00000E+00   ,     26
     C  .22339E+01, .29629E+02, .13427E+01, .00748E+00, .00000E+00   ,     27
     C  .22339E+01, .29629E+02, .13427E+01, .00748E+00, .00000E+00   ,     28
     C  .22339E+01, .29629E+02, .13427E+01, .00748E+00, .00000E+00   ,     29
     C  .22339E+01, .29629E+02, .13427E+01, .00748E+00, .00000E+00   ,     30
     C  .23581E+01, .11306E+03, .99930E+00, .01923E+00, .00000E+00   /     31
C
C          MODIFIER OMEGA COEFFICIENTS
      DATA ((GMPO(I,J),I=1,4),J=1,16)/                                     ISPG
     C   .1780E+01, -.3000E+01,  .1620E+02,  .2270E+00,                     01
     C   .7190E+00, -.1090E+02,  .1688E+04,  .3750E+00,                     02
     C   .1360E+01, -.2640E+01,  .1150E+02,  .3860E+00,                     03
     C   .5000E+01, -.1010E+01,  .3640E+01,  .0        ,                    04
     C   .1760E+01, -.1510E+01,  .2630E+01,  .2330E+00,                     05
     C   .3800E+01, -.1520E+01,  .6540E+01,  .3480E+00,                     06
     C   .1780E+01, -.3000E+01,  .1620E+02,  .2270E+00,                     07
     C   .2540E+01, -.1140E+01,  .2260E+01,  .0        ,                    08
     C   .1270E+01, -.1340E+01,  .1050E+01,  .0        ,                    09
     C   .1360E+01, -.2640E+01,  .1150E+02,  .3860E+00,                     10
     C   .5000E+01, -.5680E+00,  .1830E+01,  .6300E-01,                     11
     C   .1400E+01, -.2030E+01,  .1040E+02,  .6940E+00,                     12
     C   .1400E+01, -.2030E+01,  .1040E+02,  .6940E+00,                     13
     C   .1400E+01, -.2030E+01,  .1040E+02,  .6940E+00,                     14
     C   .5000E+01, -.9700E+00,  .4400E+01,  .2680E+00,                     15
     C   .6790E+00, -.1097E+02,  .1568E+04,  .4830E+00/                     16
      DATA ((GMPO(I,J),I=1,4),J=17,31)/
     C   .1590E+01, -.3270E+01,  .2670E+02,  .4120E+00,                     17
     C   .1170E+01, -.4590E+01,  .2919E+02,  .4300E+00,                     18
     C   .5000E+01, -.1380E+01,  .8260E+01,  .3260E+00,                     19
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        ,                    20
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        ,                    21
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        ,                    22
     C   .1660E+01, -.2620E+01,  .9970E+01,  .5150E+00,                     23
     C   .1130E+01, -.4640E+01,  .1646E+03,  .6480E+00,                     24
     C   .1080E+01, -.6600E+01,  .3461E+03,  .3950E+00,                     25
     C   .1980E+01, -.1750E+01,  .3670E+01,  .2320E+00,                     26
     C   .2310E+01, -.1670E+01,  .3940E+01,  .0        ,                    27
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        ,                    28
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        ,                    29
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        ,                    30
     C   .1980E+01, -.9740E+00,  .1640E+01,  .0        /                    31
C
C          MODIFIER BETA COEFFICIENTS
      DATA ((GMPB(I,J),I=1,2),J=1,16)/                                     ISPG
     C   .4020E+00,  .2300E+00,                                             01
     C   .2030E+01, -.3540E+00,                                             02
     C   .9700E-01,  .7550E+00,                                             03
     C   .1507E+01, -.5200E+00,                                             04
```

```
     C    .9270E+00, -.2990E+00,                                          05
     C    .5220E+00,  .1730E+00,                                          06
     C    .3900E-01,  .1000E+01,                                          07
     C    .5260E+00,  .1360E+00,                                          08
     C    .4600E-01,  .1000E+01,                                          09
     C    .9700E-01,  .7550E+00,                                          10
     C    .2600E+00,  .4190E+00,                                          11
     C    .1810E+00,  .4450E+00,                                          12
     C    .1810E+00,  .4450E+00,                                          13
     C    .1810E+00,  .4450E+00,                                          14
     C    .1000E+00,  .6290E+00,                                          15
     C    .2020E+00,  .4540E+00/                                          16
      DATA ((GMPB(I,J),I=1,2),J=17,31)/
     C    .3530E+00,  .1820E+00,                                          17
     C    .1420E+00,  .5240E+00,                                          18
     C    .4530E+00,  .3400E+00,                                          19
     C    .5100E-01,  .1000E+01,                                          20
     C    .2780E+00,  .3650E+00,                                          21
     C    .1365E+01, -.2080E+00,                                          22
     C    .2800E+00,  .2280E+00,                                          23
     C    .9300E-01,  .1000E+01,                                          24
     C    .2090E+00,  .5430E+00,                                          25
     C    .1100E+00,  .6780E+00,                                          26
     C    .4410E+00,  .1730E+00,                                          27
     C    .2780E+00,  .3650E+00,                                          28
     C    .2780E+00,  .3650E+00,                                          29
     C    .2780E+00,  .3650E+00,                                          30
     C    .2780E+00,  .3650E+00/                                          31
C
C           MODIFIER CHECK COEFFICIENTS
      DATA ((GMPC(I,J),I=1,2),J=1,16)/                                   ISPG
     C    1.     E 00,   .225 E 03,                                       01
     C    1.     E 00,   .300 E 03,                                       02
     C    1.     E 00,   .300 E 03,                                       03
     C    1.     E 00,   .350 E 03,                                       04
     C    1.     E 00,   .325 E 03,                                       05
     C    1.     E 00,   .300 E 03,                                       06
     C    1.     E 00,   .250 E 03,                                       07
     C    1.     E 00,   .350 E 03,                                       08
     C    1.     E 00,   .300 E 03,                                       09
     C    1.     E 00,   .300 E 03,                                       10
     C    1.     E 00,   .250 E 03,                                       11
     C    1.     E 00,   .250 E 03,                                       12
     C    1.     E 00,   .250 E 03,                                       13
     C    1.     E 00,   .250 E 03,                                       14
     C    1.     E 00,   .250 E 03,                                       15
     C    1.     E 00,   .250 E 03/                                       16
      DATA ((GMPC(I,J),I=1,2),J=17,31)/
     C    1.     E 00,   .250 E 03,                                       17
     C    1.     E 00,   .250 E 03,                                       18
     C    1.     E 00,   .250 E 03,                                       19
     C    1.     E 00,   .250 E 03,                                       20
     C    1.     E 00,   .275 E 03,                                       21
     C    1.     E 00,   .250 E 03,                                       22
```

```
C    1.     E 00,   .250 E 03,
C    1.     E 00,   .250 E 03,
C    1.     E 00,   .250 E 03,
C    1.     E 00,   .275 E 03,
C    1.     E 00,   .350 E 03,
C    1.     E 00,   .275 E 03,
C    1.     E 00,   .275 E 03,
C    1.     E 00,   .275 E 03,
C    1.     E 00,   .275 E 03/
  END
```

Brand, Gary J.

1981. "Grow"—a computer subroutine that projects the growth of trees in Lake States' forests. U.S. Department of Agriculture Forest Service, Research Paper NC-207, 11 p. U.S. Department of Agriculture Forest Service, North Central Forest Experiment Station, St. Paul, Minnesota.

A computer subroutine, Grow, has been written in 1977 Standard FORTRAN to implement a distance-independent, individual tree growth model for Lake States' forests. Grow is a small and easy-to-use version of the growth model. All the user has to do is write a calling program to read initial conditions, call Grow, and summarize the results.

KEY WORDS: Growth model, mortality, FORTRAN.

☆ U.S. Government Printing Office: 1981—767-088/212 Region No. 6